


凌启渝 编著

BASIC万花筒



科学普及出版社

BASIC 万花筒

——适用Apple机及其兼容机

凌启渝 编著

科学普及出版社

内 容 提 要

本书利用Apple(苹果)型微电脑的多种性能,灵活地设计了一系列程序,使电脑除了解复杂的数学题外,还能教英语、弹钢琴、画彩色图画、猜谜、作数学乌龟;本书还为生活中处处可见的数学问题编制了程序,大大启发读者的思路。本书程序简单,叙述清楚,操作容易,很适合青少年电脑初学者学习。

BASIC 万花筒

——适用 Apple 机及其兼容机

凌 启 渝 编 著

责任编辑: 宋宜昌

科学普及出版社出版(北京海淀区白石桥路32号)
新华书店北京发行所发行 各地新华书店经售
北京丰华印刷厂印刷

开本: 787×1092毫米1/32印张: 3.125 字数: 40 千字

1987年2月第1版 1987年2月第1次印刷

印数: 1—4,500册 定价: 0.60元

统一书号: 7051·1032 本社书号: 1155

目 录

1. 让计算机帮你练习打字	1
2. 计算机老师	4
3. 光电小钢琴	7
4. 在屏幕上画图	15
5. 画一幅海景	23
6. 猜日期	26
7. 黑盒子里是什么	29
8. 营养师	32
9. 四个整数	35
10. 接待“随便顾客”	38
11. 加密和解密	40
12. 密码	41
13. 排单词	43
14. 黑板擦	47
15. 数学乌龟	49
16. 填除法直式	55
17. 巧合	58
18. 精打细算	61
19. “三角牌”机器	63
20. 开高球	65
21. 重复数	67
22. 回文数	73

23. 多子多孙.....	7
24. 十位数是奇数.....	81
25. 16的倍数.....	84
26. 几率分布.....	86
27. 一百环.....	89
28. 过小溪.....	91
29. 曼哈顿半岛的价值.....	94

1. 让计算机帮你练习打字

你可以让计算机教你学数学，学外语，也可以让它帮你练习英文打字。当然，你得先教它。

输入以下的程序：

```
10 INPUT "WHAT CHARACTERS DO YOU WANT?";  
   C$  
20 L=LEN (C$) : DIM A(L)  
30 PRINT  
40 INPUT "HOW MANY IN EACH WORD?"; WL  
50 INPUT "DO YOU WANT ECHO?"; D$  
60 IF D$="YES" THEN E=1  
70 INPUT "HOW MANY TRIALS?"; NT  
80 FOR T=1 TO NT  
90 NP=NP+WL  
100 FOR I=1 TO WL  
110 R=INT (L * RND (5) +1)  
120 A$=A$+MID$ (C$, R, 1)  
130 NEXT I  
140 PRINT : PRINT A$  
150 FOR I=1 TO WL  
160 T$=MID$ (A$, I, 1)  
170 GET S$  
180 IF E=0 THEN 200
```

```

190 PRINT S$;
200 IF S$ <    > T$ THEN 350
210 NR = NR + 1
220 NEXT I
230 A$ = "  "
240 IF E = 1 THEN PRINT
250 NEXT T
260 PRINT : PRINT "SCORE : "; INT (100 * NR /
    NP); "%"
270 IF NR = NP THEN 310
280 PRINT "ERROR:" : FOR I = 1 TO L : IF A(I) = 0
    THEN 300
290 PRINT MID$ (C$, I, 1); " "A(I)
300 NEXT I
310 PRINT : INPUT "AGAIN?"; E$
320 IF E$ = "YES" THEN RUN
330 PRINT "GOOD BYE."
340 END
350 FOR J = 2 TO L
360 IF MID$ (C$, J, 1) <    > MID$ (A$, I, 1)
    THEN NEXT : GOTO 380
370 A(J) = A(J) + 1
380 PRINT : PRINT "ERROR ON"; MID$ (A$, I, 1)
390 A$ = "  "
400 FOR I = 1 TO 300 : NEXT
410 GOTO 250

```

下面是一次执行的结果，我们把它“翻译”一下，你就知

道该怎么使用了。

计算机问：你要练习哪几个键？

操作者答：ASDFJKL；（见执行后第一行）

计算机问：你的一个词中有几个字符？

操作者答：5（见执行后第二行）

计算机问：要不要把你打的字在屏幕上显出来？

操作者答：要（见执行后第三行）

计算机问：你试几次？

操作者答：3（见执行后第四行）

RUN

WHAT CHARACTERS DO YOU WANT? ASDFJKL；

HOW MANY IN EACH WORD?5

DO YOU WANT ECHO? YES

HOW MANY TRIALS? 3

KDS; F

KDS; J

ERROR ON F

; KDAJ

; KDAJ

ADDDL

ADDDS

ERROR ON L

SCORE: 86%

ERROR :

F 1

L 1

AGAIN? NO

GOOD BYE.

操作者作了上述选择后，计算机就从操作者要练习的键中，按规定的字符个数随机地选择字符，让他练习。比如，计算机先显示出“KDS;F”这五个字符之后，操作者打键盘，要把它们重复一遍，结果，打错了最后一个字符（应是F，打了J），计算机马上指出，在应该是F的位置上打错了字符（显示ERROR ON F）。以上算是一次练习。进行三次练习以后，计算机就统计出打字的准确率，这里是86%，并报告：在F上错一下，L上错一下。

这时，屏幕上又显示“再练吗？”(AGAIN?)如选择“YES”，即重来；“NO”，就结束。对了，计算机可不会忘记说：“再见”(GOOD-BYE)。

2. 计算机老师

要是你的英语老师教完新课，要让你默写单词，而完全不必亲自出马，可以让一台计算机来代替自己。使用下面这个“计算机辅导”程序，把要默写的单词事先放入DATA语句（程序中的30、40句，如果嫌太少还能增加）中，计算机屏幕上就会逐个出现单词。

不过，每个单词在屏幕上是一闪而过，你只能看上一眼，勉强知道要默写的是哪个单词。过后，屏幕上就出现问号，要你来默写了。你击键进行输入。拼写对的，屏幕上出现一个笑脸，计算机老师对你挺满意。拼写错了，屏幕上出现一张哭脸，计算机老师皱眉头了！不过他还是耐心地告诉你，应该怎样拼写THE WORD WHOULD BE……）。

一个单词默完，不论你对不对，计算机老师总要问一

句：“要不要再默写？”(AGAIN?)如果你答“要”(YES)，屏幕上再来一个单词；要是你累了，想休息，可以回答“不要”(NO)，计算机老师就对你说“再见”(GOOD-BYE)，放你去玩。

事实上，因为我们的判断语句(260句)只根据“输入的字符串C\$是不是“YES”来判断，所以，只要你的回答不是“YES”(如是“NO”，“YS”)，计算机老师就停止工作。

要是插入下面几句语句，计算机老师还会为你统计默写了多少个单词，其中拼对了多少，占百分之几。

```
22 S=S+1
```

```
172 S1=S1+1
```

```
265 PRINT "YOUR SCORE:", S1, "/", S, "=",  
S1/S * 100, "%"
```

计算机老师的程序

```
10 REM TUTORING
```

```
20 READ A$
```

```
25 PRINT A$
```

```
30 DATA DETERMINATION, A, B, C, D
```

```
40 DATA E, F, G, H, I, STOP
```

```
50 IF A$="STOP" THEN 230
```

```
60 FOR I=1 TO 25
```

```
62 PRINT
```

```
64 FOR J=1 TO 20: NEXT J
```

```
66 NEXT I
```

```
70 INPUT B$
```

```

80 PRINT TAB( 23)“ * * * * * ”
90 PRINT TAB( 22)“ * ”
100 PRINT TAB( 21)“ * - - ”
110 PRINT TAB( 20)“ * 0 0 ”
120 PRINT TAB( 20)“ * # ”
130 IF A$ < > B$ THEN 190
140 PRINT TAB( 20)“ * * * ”
150 PRINT TAB( 21)“ * * * ”
160 PRINT TAB( 22)“ * * * * * ”
170 PRINT TAB( 23)“ * * * * * ”
175 PRINT : PRINT
180 GOTO 250
190 PRINT TAB( 20)“ * * * * * ”
200 PRINT TAB( 21)“ * ( ) ”
210 PRINT TAB( 22)“ * ”
220 PRINT TAB( 23)“ * * * * * ”
225 PRINT : PRINT
230 PRINT “THE WORD SHOULD BE ”, A$, “. ”
240 PRINT : PRINT
250 INPUT “AGAIN?”; C$
255 PRINT : PRINT : PRINT
260 IF C$ = “YES” THEN 20
270 PRINT “GOOD-BYE!”
280 END

```


执 行 结 果

←RUN

要默的词 → DETERMINATION

学生默的→?DETERMINATION

对了→



再来呢? AGAIN? YES


是

要默的词→A

这个A是假设一个单词

学生默的→?B

错了→



应该是 A→THE WORD SHOULD BE A

再来呢? AGAIN?N

不了

再见! GOOD-BYE!

3. 光电小钢琴

苹果机是会演奏乐曲的，当然，你得先编程序“教”它。

试一试下面的程序：

```
10 FOR J=1 TO 10
20 X=PEEK (-16336)
30 NEXT J
40 END
```

就能听到粗糙的声音。这不稀奇，苹果机可以发出太空时代“激光枪”的声音，会学鸭叫、蜂鸣，也可以奏出悦耳动听乐曲。用 BASIC 做这件事速度太慢，必须用机器语言子程序，这个子程序夹在 BASIC 程序当中，需要的时候就调用。

让我们先讲一讲机器语言发声程序。击键输入如下的程序，它是作为子程序用的，最后有 RETURN。行号从 1000 开始，有人习惯这样写。

```
1000 FOR ADDRESS=770 TO 790
1020 READ COMM
1030 POKE ADDRESS, COMM
1040 NEXT ADDRESS
1050 DATA 173, 48, 192, 136, 208, 5, 206, 1, 3
1060 DATA 240, 9, 202, 208, 245, 174, 0, 3, 76, 2, 3, 96
1090 RETURN
```

这个子程序直接向指定的存贮单元存数，这是用 BASIC 程序让机器码进入苹果机的一种方式。

1020 句从 DATA 语句中读入数据。

1030 句把 DATA 语句中的数据一个接一个地存入相应的一串存贮单元（地址是 770—790）中去。循环的第一步，地址为 770。这时，第一个数据 173 被读进，并放进 770 单元。循环的第二步，地址码变为 771，第二个数据 48 被读了进去……

直到最后一个数据(96) 被读进最后一个启用的存储单元(790)中。

在把数据放入存贮单元中去时,要十分小心,要挑“空闲”的用,否则够麻烦的。一般说来,768~999单元是空闲区域,可以放心使用。

那么,放置进去的173、48……之类数据是什么意思呢?对苹果机来说,它们就象是BASIC中的GOTO、GR之类的指令。我们现在暂且不去管它。

现在你可以键入主程序了,这个主程序走到30句,进入刚才的机器语言子程序,将特定的数据放入770~790各存贮单元。这个过程只要执行一次就够了,从90句返回时回到40句,

```
10 REM MAKING SOUNDS
```

```
20 HOME
```

```
30 GOSUB 1000
```

```
40 INPUT "FREQUENCY- 1 TO 255 : "; F
```

```
50 POKE 768, F
```

```
60 POKE 769, 255
```

```
70 CALL 770
```

```
80 PRINT
```

```
90 GOTO 40
```

```
100 END
```

40句要求你键入从1~255之间的任一数据F。50句把F送入存贮单元768中去。当机器语言子程序被调用时,苹果机会从存贮单元768中找到它发出的声音的频率。

60句把255放入存贮单元769中,这个数据决定发声的长短,要是你希望声音短促一些,可以把这个数据改小一点。

70句把从存贮单元770开始的机器语言子程序调用出来，就是我们刚才放好的那一段。这样，就产生了一个频率为F、长度为255的声音。

那么F的值如何选取呢？你如果按192，听到的是中央C；按53，听到的正好是出错铃。当然，这一次并不一定表示有错。

好，把上面两段程序并在一起，执行一遍，就可以奏乐。不过，一定要小心翼翼，输入A\$时也挺慢，根本不能连贯起来。请把上面的程序保留好，下面我们对它进行改进。

首先，作为乐器要能演奏音阶，按一个键就发一个音。

我们保留上面1000—1100的子程序，要使八个基本键（打字机上的A,S,D,F,J,K,L和,）分别代表音阶中的C、D、E、F、G、A、B、C音。

基本的想法是：

屏幕清零，执行发声程序（20、40句），

从键盘上得到一个字母（50句），

将该字母翻译成特定频率（70—140句）

在屏幕上显示这个音的音名（要是你按的键不是这八个基本键之一，计算机按照前一个音再发一次，这是约定）

按照这个频率发音（250—270句），

重复上述过程，让你继续弹奏（300句）。

程序如下：

```
10 REM A LITTLE PIANO
```

```
20 HOME
```

```
30 REM LOAD THE MACHINE LANGUAGE SUB-  
ROUTINE INTO MEMORY
```

```
40 GOSUB 1000
```

```

50 GET A$
60 VTAB 20 : REM FOR POSITIONING NOTES ON
  SCREEN
65 REM TRANSLATE KEYBOARD KEY INTO A
  NOTE
70 IF A$="A" THEN F=192 : PRINT "C"
80 IF A$="S" THEN F=171 : PRINT "D"
90 IF A$="D" THEN F=152 : PRINT "E"
100 IF A$="F" THEN F=144 : PRINT "F"
110 IF A$="J" THEN F=128 : PRINT "G"
120 IF A$="K" THEN F=114 : PRINT "A"
130 IF A$="L" THEN F=101 : PRINT "B"
140 IF A$=";" THEN F=95 : PRINT "C"
150 IF A$="Q" THEN 310
250 POKE 768,F : REM FREQUENCY
260 POKE 769, 100 : REM DURATION
280 CALL 770 : REM CALLING MACHINE LAN-
  GUAGE SUBROUTINE TO PRODUCE NOTE
300 GOTO 50 : REM NFXT NOTE
310 END
1000 FOR ADDRESS=770 TO 790
1020 READ COMM
1030 POKE ADDRESS, COMM
1040 NEXT ADDRESS
1050 DATA 173,48,192,136,208,5,206,1,3
1060 DATA 240,9,202,208,245,174,0,3,76,2,3,96
1090 RETURN

```


下面给你一张音名和与之对应的声音频率表，你编制音乐程序时可以方便些：

#A	215	#A	107	#A	53
B	203	B	101	B	50
C	192	C	95	C	47
#C	181	#C	90	#C	45
D	171	D	85	#D	42
#D	161	#D	80	#D	40
E	152	E	75	E	38
F	144	F	71	F	35
#F	136	#F	67		
G	127	G	63	G	33
#G	121	#G	60		
A	114	A	56	A	31

要是你想演奏《星星闪烁》，就要按

AAJJKKJ

FFDDSSA

JJFFDDSS

JJFFDDSS

AAJJKKJ

FFDDSSA

用唱名法表示是：

1155665

4433221

5544332

5544332

1155665